Real's HowTo PDF version.

This is the PDF version of the Real's HowTo Web site (http://www.rgagnon.com/howto.html).

For up−to−date content, please refer to the Web site.

There are 3 files : Real's Java , Real's Javascript and Real's Powerbuilder HowTo.

Please don't make PDF versions available on the internet (it's ok in intranet)

From the PDF, you can't run the exemples and the links to other How−to's are not working.

If you feel that effort has been useful to you, perhaps you will consider giving something back?

You can make a donation through PayPal at https://www.paypal.com , make you donation to **real@rgagnon.com**

Contributions via PayPal are accepted in any amount using a credit card or checking account.

(Donations of **any size** gladly accepted)

Real's HowTo@www.rgagnon.com

1 WSH VBScript

1.1 wsh−vbs

1.2 Read me

These How−to's are about the usage of VBScript used with the Window Scripting Host (WSH).

WSH enables scripts to be executed directly on the Windows desktop or command console, without the need to embed those scripts in an HTML document. Scripts can be run directly from the desktop simply by clicking on a script file (.vbs for VBScript), or from the command console. The WSHost is included with all recent Windows versions or it can be freely downloaded from Microsoft or here

1.3 Send email with Outlook (MAPI)

```
[sendemail.vbs]
Dim ToAddress
Dim MessageSubject
Dim MessageBody
Dim MessageAttachment

Dim ol, ns, newMail

ToAddress = "Gagnon, Réal"   ' change this...
MessageSubject = "VBS MAPI HowTo"
MessageBody = "*BODY* email via MAPI *BODY*"

Set ol = WScript.CreateObject("Outlook.Application")
Set ns = ol.getNamespace("MAPI")
ns.logon "","",true,false
Set newMail = ol.CreateItem(olMailItem)
newMail.Subject = MessageSubject
newMail.Body = MessageBody &vbCrLf

' validate the recipient, just in case...
Set myRecipient = ns.CreateRecipient(ToAddress)
myRecipient.Resolve
If Not myRecipient.Resolved Then
   MsgBox "unknown recipient"
Else
   newMail.Recipients.Add(myRecipient)
   newMail.Send
End If

Set ol = Nothing
```

1.4 Send email with attachment with Outlook (MAPI)

```
[Email2Elvis.vbs]
```

```
Dim ToAddress
Dim MessageSubject
Dim MessageBody
Dim MessageAttachment

Dim ol, ns, newMail

ToAddress = "Presley, Elvis"
MessageSubject = "Don't be cruel"
MessageBody = "Tutti Frutti"
MessageAttachment = "d:\work\report.txt"

' connect to Outlook
Set ol = WScript.CreateObject("Outlook.Application")
Set ns = ol.getNamespace("MAPI")

Set newMail = ol.CreateItem(olMailItem)
newMail.Subject = MessageSubject
newMail.Body = MessageBody &vbCrLf

' validate the recipient, just in case...
Set myRecipient = ns.CreateRecipient(ToAddress)
myRecipient.Resolve
If Not myRecipient.Resolved Then
   MsgBox "Unknown recipient"
Else
   newMail.Recipents.Add(ToAddress)
   newMail.Attachments.Add(MessageAttachment).Displayname = "Check this out"
   newMail.Send
End If

Set ol = Nothing
```

1.5 Write to a file

```
[writefile.vbs]

Dim objFileSystem, objOutputFile
Dim strOutputFile

' generate a filename base on the script name
strOutputFile = "./" &Split(WScript.ScriptName, ".")(0) &".out"

Set objFileSystem = CreateObject("Scripting.fileSystemObject")
Set objOutputFile = objFileSystem.CreateTextFile(strOutputFile, TRUE)

objOutputFile.WriteLine("Hello world (" &Now &")")
objOutputFile.Close

Set objFileSystem = Nothing

WScript.Quit(0)
```

To append to an existing file :

```
[appendfile.vbs]

Dim objFileSystem, objOutputFile
Dim strOutputFile

Const OPEN_FILE_FOR_APPENDING = 8
```

```
' generate a filename base on the script name
strOutputFile = "./writefile.out"

Set objFileSystem = CreateObject("Scripting.fileSystemObject")
Set objOutputFile = objFileSystem.OpenTextFile(strOutputFile, OPEN_FILE_FOR_APPENDING)

objOutputFile.WriteLine("Hello world (" &Now &")")
objOutputFile.Close

Set objFileSystem = Nothing

WScript.Quit(0)
```

1.6 Read a file

```
[readfile.vbs]

Dim objFileSystem, objInputFile
Dim strInputFile, inputData, strData

Const OPEN_FILE_FOR_READING = 1

' generate a filename base on the script name, here readfile.in
strOutputFile = "./" &Split(WScript.ScriptName, ".")(0) &".in"

Set objFileSystem = CreateObject("Scripting.fileSystemObject")
Set objInputFile = objFileSystem.OpenTextFile(strOutputFile, OPEN_FILE_FOR_READING)

' read everything in an array
inputData = Split(objInputFile.ReadAll, vbNewline)

For each strData In inputData
    WScript.Echo strData
Next

objInputFile.Close
Set objFileSystem = Nothing

WScript.Quit(0)
```

1.7 Access arguments on the command line

```
[args.vbs]

Dim arg

If WScript.Arguments.Count = 0 Then
   WScript.Echo  "no argument on the command line."
Else
   For each arg in WScript.Arguments
      WScript.Echo "arg : "  &arg
   Next
End If

WScript.Quit(0)
```

1.8 Copy, delete or move a file

```
' copy
Set fso = CreateObject("Scripting.FileSystemObject")
Set aFile = fso.CreateTextFile(".\output.dat", True)
aFile.WriteLine("1234")
Set aFile = fso.GetFile(".\output.dat")
aFile.Copy("./output.bak")

' alternate
fso.CopyFile "c:\mydir\*.*", "d:\backup\",TRUE

' copy folder
fso.CopyFolder "c:\mydir\*", "D:\backup\mydir\",TRUE

' delete
Set fso = CreateObject("Scripting.FileSystemObject")
Set aFile = fso.GetFile(".\output.dat")
aFile.Delete
' move or rename
Set fso = CreateObject("Scripting.FileSystemObject")
Set aFile = fso.CreateTextFile(".\output.dat", True)
aFile.WriteLine("1234")
Set aFile = fso.GetFile(".\output.dat")
aFile.Move ".\output.ok"
```

1.9 Deal with environment variable or registry key

The WSH provides a an object, WShell, to manipulate the environment.

```
[env.vbs]

Dim WSHShell
Set WSHShell = WScript.CreateObject("WScript.Shell")

WScript.Echo "The current PATH is " &_
    WSHShell.Environment.item("path")

WScript.Echo "Creating a new environment variable called RealHome"
' this variable will exist only during execution
' time of this script (Win95)
WSHShell.Environment.item("RealHome") = "Hello world"
WScript.Echo "Realhome is " &_
    WSHShell.Environment.item("RealHome")

Set WSHShell = Nothing
WScript.Quit(0)
[reg.vbs]

Dim WSHShell
Set WSHShell = WScript.CreateObject("WScript.Shell")

' write in the HKey_Current_User
WSHShell.RegWrite "HKCU\RealHome\", "Welcome"

WSHShell.RegWrite "HKCU\RealHome\How-to", "Java"
WSHShell.RegWrite "HKCU\RealHome\How-to", "Javascript"
WSHShell.RegWrite "HKCU\RealHome\How-to", "PB"
```

```
WSHShell.RegWrite "HKCU\RealHome\How-to", "VBScript"

WScript.Echo "Value of HKCU\Realhome is " &_
    WSHShell.RegRead("HKCU\RealHome\")

Set WSHShell = Nothing
WScript.Quit(0)
```

1.10 Execute external programs

```
[run.vbs]

Dim WSHShell
Set WSHShell = WScript.CreateObject("WScript.Shell")

' open maximized and wait
WSHShell.Run "notepad.exe", 3, true
' open minimized and wait
WSHShell.Run "notepad.exe", 2, true
' open normal and don't wait
WSHShell.Run "notepad.exe", 1, false

Set WSHShell = Nothing
WScript.Quit(0)
```

1.11 Create and use a class

```
[myclass.vbs]

Dim my_howto

' create an VbsHowTo object and init its properties
Set my_howto = new VbsHowTo
my_howto.create "General", _
                "Start a script", _
                "CSCRIPT to start a script in CONSOLE mode" &vbCRLF &_
                "WSCRIPT to start a script in GUI mode"
my_howto.print
WScript.Quit(0)

' ----------------------------------------
' class definition to represent an VBScript Howto
Class VbsHowTo
 Dim category
 Dim title
 Dim howto

 Sub create(cat, tit, how)
    category = cat
    title = tit
    howto = how
 End Sub

 Sub print()
   WScript.Echo category &" - " &title
   Wscript.Echo howto
 End Sub
```

```
End Class
```
Thanks to E. Girardet for this tip

1.12 Call IE

```
[iewithurl.vbs]

' Display a URL
Dim IE
Set IE = CreateObject("InternetExplorer.Application")

With IE
        .left=200
        .top=200
        .height=400
        .width=400
        .menubar=0
        .toolbar=1
        .statusBar=0
        .navigate "http://tactika.com/realhome/realhome.html"
        .visible=1
End With

'wait a while until IE as finished to load
Do while IE.busy
loop

Set IE = Nothing

WScript.Quit(0)
```

```
[iewithhtml.vbs]
' start IE, and create HTML on-the-fly

Dim IE
Set IE = CreateObject("InternetExplorer.Application")

With IE
        .left=200
        .top=200
        .height=140
        .width=250
        .menubar=0
        .toolbar=0
        .statusBar=0
        .navigate "About:Blank"
        .visible=1
End With

'wait a while until IE as finished to load
Do while IE.busy
loop

With IE.document
        .Open
        .WriteLn "<HTML><HEAD>"
        .WriteLn "<TITLE>HELLO!</TITLE></HEAD>"
        .WriteLn "<BODY>"
        .WriteLn "<b>VBScript says </b> Hello world"
        .WriteLn "</BODY>"
```

```
        .WriteLn "</HTML>"
        .Close
End With

Set IE = Nothing
WScript.Quit(0)
```

Thanks to E. Girardet for this tip

1.13 Connect to a database

```
[odbcselect.vbs]
Dim OdbcDSN
Dim connect, sql, resultSet

OdbcDSN = "DSN=Sybase Demo DB V6 DWB;UID=dba;PWD=sql"
Set connect = CreateObject("ADODB.Connection")
connect.Open OdbcDSN

sql="SELECT emp_fname, emp_lname FROM employee"

Set resultSet = connect.Execute(sql)
On Error Resume Next
resultSet.MoveFirst
Do While Not resultSet.eof
  WScript.Echo resultSet("emp_lname") &" , " &resultSet("emp_fname")
  resultSet.MoveNext
Loop

resultSet.Close
connect.Close
Set connect = Nothing

WScript.Quit(0)
```

```
[odbcupdate.vbs]
'---- CursorTypeEnum Values ----
Const adOpenForwardOnly = 0
Const adOpenKeyset = 1
Const adOpenDynamic = 2
Const adOpenStatic = 3

'---- LockTypeEnum Values ----
Const adLockReadOnly = 1
Const adLockPessimistic = 2
Const adLockOptimistic = 3
Const adLockBatchOptimistic = 4

Dim OdbcDSN
Dim connect, resultSet

OdbcDSN = "DSN=Sybase Demo DB V6 DWB;UID=dba;PWD=sql"
Set connect = CreateObject("ADODB.Connection")
connect.Open OdbcDSN

Set resultSet = CreateObject("ADODB.Recordset")
Set resultSet.ActiveConnection = connect
resultSet.Source = "SELECT dept_id, dept_name, dept_head_id FROM department WHERE 1=2"
resultSet.CursorType = adOpenStatic
resultSet.LockType = adLockOptimistic
resultSet.Open
```

```
On Error Resume Next

resultSet.Addnew
resultSet("dept_id") = 1234
resultSet("dept_name") = "VBScript How-to"
resultSet("dept_head_id") = 501

resultSet.Update

resultSet.Close
connect.Close
Set resultSet = Nothing
Set connect = Nothing

WScript.Quit(0)
```

Thanks to E. Girardet for this tip

1.14 Detect files changed since "n" days

```
[filechange.vbs]

Function FilesModifiedSince (FolderSpec, Days)
  Dim fso, fc, f, d
  Set fso = CreateObject("Scripting.FileSystemObject")
  Set fc = fso.GetFolder(FolderSpec).Files
  Set d = CreateObject("Scripting.Dictionary")

  For Each f in fc
    If DateDiff("d", f.DateLastModified, Now) <= Days Then d.Add f, f.DateLastModified
  Next
  Set fso = Nothing
  Set fc = Nothing
  Set FilesModifiedSince = d
End function

' Example, get all the files modifed since 5 days
' in the specified directory
Dim a, f

Set f  = FilesModifiedSince("c:\dev\work\scripting", 5)
a = f.keys
b = f.items

For i = 0 To f.count - 1
  WScript.Echo  a(i) &"   " &b(i)
Next

Set f = Nothing
WScript.Quit(0)
```

1.15 Detect WSH, Script engine and DLL versions

```
Dim WSHShell, FSO, VbDll, VbDllVersion, s

Set WSHShell = WScript.CreateObject("WScript.Shell")
WScript.Echo "WSH version : " &WScript.Version
```

```
s = ""
s = ScriptEngine &" Version "
s = s &ScriptEngineMajorVersion &"."
s = s &ScriptEngineMinorVersion &"."
s = s &ScriptEngineBuildVersion
WScript.Echo "Script engine version : " &s

Set FSO = WScript.CreateObject("Scripting.FileSystemObject")
If WSHShell.Environment.item("OS") = "" Then
   ' we assume a non-NT system
   VbDll = "c:\windows\system\vbscript.dll"
Else
   VbDll = "c:\winnt\system32\vbscript.dll"
End If

VbDllVersion = FSO.GetFileVersion(VbDll)
WScript.Echo "VBScript Dll version : " &VbDllVersion &" (" &VbDll &") "

Set WSHShell = Nothing
Set FSO = Nothing
WScript.Quit(0)
```

1.16 Print a document

```
[printdoc.vbs]

' Word need to be installed on this machine
Dim DocToPrint, oWord, oDoc
DocToPrint = "C:\MyFile.Doc"
Set oWord = CreateObject("Word.Application")
Set oWordActiveDoc = oWord.Documents.Open("" &DocToPrint)
oWordActiveDoc.PrintOut

Set oWord = Nothing
WScript.Quit(0)
```

1.17 Reboot a machine

```
[reboot.vbs]

Dim WSHShell
Set WSHShell = WScript.CreateObject("WScript.Shell")

' win9x
WSHShell.Run _
  "C:\WINDOWS\RUNDLL32.EXE shell32.dll,SHExitWindowsEx 2", 2, false

' on NT, you can simply call REBOOT.EXE utility
' which is included with the NT resource KIT
WScript.Quit(0)
```

1.18 Tokenize a string

```
[tokenize.vbs]
```

```
Dim  tokens
tokens = Split("This is the string to be tokenized", " ")

For i=0 To UBound(tokens)
   WScript.Echo tokens(i)
Next

WScript.Quit(0)
```

1.19 Capture the output of an external program

```
[capture.vbs]
Const TemporaryFolder = 2

strDOSCmd = "dir c:\*.txt /b /on"
Set WSHShell = CreateObject("Wscript.Shell")
Set fso = CreateObject("Scripting.FileSystemObject")
tempfile = fso.GetTempName
path = fso.GetSpecialFolder(TemporaryFolder)
tempfile = path &"\" &tempfile
WSHShell.Run _
   "%comspec% /c " &strDOSCmd &" >" &tempfile, 0, true
arResults = Split(fso.OpenTextFile(tempfile).ReadAll,vbcrlf)
fso.DeleteFile tempfile
WScript.Echo join(arResults,vbcrlf)
WScript.Quit(0)
```

1.20 Find a file

```
[ff.vbs]
Dim System
Dim Drive
ReDim Location(0)
Dim FileName
FileName = LCase(Trim(InputBox("Enter the name of the" _                  " file that you wish
If Len(FileName) = 0 Then Wscript.Quit
Set System = CreateObject("Scripting.FileSystemObject")
For Each Drive In System.Drives
   If Drive.IsReady And Drive.DriveType = 2 Then
      Call FindFile(Drive &"\")
   End If
Next 'Drive
Msgbox Join(Location,vbCr)


Sub FindFile(ThisFolder)
    Dim File
    Dim Folder
    For Each Folder In System.GetFolder(ThisFolder).SubFolders
       For Each File In Folder.Files
          If LCase(File.Name) = FileName Then
             Location(Ubound(Location)) = File
             ReDim Preserve Location(Ubound(Location) + 1)
          End If
       Next 'File
       Call FindFile(Folder)
    Next 'Folder
```

```
End Sub
```

Be patient, this script is not very fast if you a lot files or directory to scan.

1.21 Detect if a program is running

```
Set WshShell = WScript.CreateObject("WScript.Shell")

Set locator = CreateObject("WbemScripting.SWbemLocator")
Set service = locator.ConnectServer()
Set props = service.ExecQuery("select name, description from Win32_Process where name = 'progra
num = props.Count

If num > 0 Then
   'Program is running
Else
   'Program is not running
End If

Set WSHShell = Nothing
```

1.22 Change the current directory

```
' display the current directory
WScript.Echo WshShell.CurrentDirectory

' set the current directory
WshShell.CurrentDirectory = "c:\work"
```

1.23 Check for new email in Outlook

```
Set otl = createobject("outlook.application")
Set session = otl.getnamespace("mapi")
session.logon
Set inbox = session.getdefaultfolder(6)
c = 0
For Each m In inbox.items
  If m.unread Then c = c + 1
Next
session.logoff
s = "s"
If c = 1 Then s = ""
Msgbox "You have " &c &" unread message" &s
```

Alternate method

```
    Dim onMAPI As NameSpace
    Dim ofFolder As MAPIFolder
    Dim oItem As Object
    Dim omNewMail As MailItem

    Set oWSHShell = CreateObject("wscript.shell")

    Set onMAPI = GetNamespace("MAPI")
    Set ofFolder = onMAPI.GetDefaultFolder(olFolderInbox)
```

```
    Set omNewMail = ofFolder.Items.GetFirst

    Ret = oWSHShell.PopUp("You have new mail from " &_
              omNewMail.SenderName &_
              " about " &omNewMail.Subject &"." &vbCrLf &_
              "Would you like to read it now?", 5, _
              "You have New Mail ", 65)

    If Ret = vbYes Then
        omNewMail.Display
    End If
```

1.24 Check if a service is running

```
' W2K
Set sh = CreateObject("Shell.Application")
MsgBox sh.IsServiceRunning("Alerter")
```

1.25 Get an exit code from a vbs

Script returning an exit code [script.vbs]

```
sExitCode = InputBox ( "Enter Exit Code (0 - 255)", "Script", "0")
If Not IsNumeric(sExitCode) Then sExitCode = 0
WScript.Quit(sExitCode Mod 255)
```

Script to handle the return code

```
Set oWS = WScript.CreateObject("WScript.Shell")
sReturnCode = oWS.Run("wscript.exe script.vbs", 0, True)
MsgBox "Script2's Return Code: " &sReturnCode
```

1.26 Get the "Program Files" path

```
Set oShell = CreateObject("WScript.Shell")
Set oProcEnv = oShell.Environment("PROCESS")

sProgramFiles = oProcEnv("ProgramFiles")

If sProgramFiles = "" Then
  sProgramFiles = oShell.RegRead _
     ("HKLM\SOFTWARE\Microsoft\Windows\CurrentVersion\ProgramFilesDir")
End If

WScript.Echo sProgramFiles
```

1.27 Download HTML page

```
With CreateObject("MSXML2.XMLHTTP")
  .open "GET", "http://www.rgagnon.com/howto.html", False
  .send
```

```
  WScript.Echo .responseText
End With

responseText Property
```

If you need to send an username and password :

```
With CreateObject("MSXML2.XMLHTTP")
  .open "GET", "http://www.rgagnon.com/secure/example.html", False, _
      myusername, mypassword
  .send
  WScript.Echo .responseText
End With

responseText Property
```

1.28 Extract data from HTML page

```
theURL = "www.rgagnon.com/masters/wsh-vbs.html"

with CreateObject("InternetExplorer.Application")
  .Navigate("http://" & theURL)
  Do until .ReadyState = 4
     WScript.Sleep 50
  Loop
  With .document
    set theTables = .all.tags("table")
    nTables = theTables.length
    for each table in theTables
      s = s & table.rows(0).cells(0).innerText _
        & vbNewLine &vbNewLine
    next
    wsh.echo "Number of tables:", nTables, vbNewline
    wsh.echo "First table first cell:", s
    ' get the data with an ID
    ' msgbox ie.document.getelementbyid("d1").innerHtml
  End With
End With
```

Output is :

```
>cscript ieextract.vbs
Microsoft (R) Windows Script Host Version 5.6
Copyright (C) Microsoft Corporation 1996-2001. Tous droits réservés.

Number of tables: 1

First table first cell: VBScript
```

1.29 Create unique folder name

```
Option Explicit

Dim objFSO, dNow, yy, mt, dd, hh, nn, ss, NewFolder
Set objFSO = CreateObject("Scripting.FileSystemObject")

dNow = Now
yy = Right(Year(dNow), 2)
mt = Right("00" &Month(dNow), 2)
dd = Right("00" &Day(dNow), 2)
```

```
hh = Right("00" &Hour(dNow), 2)
nn = Right("00" &Minute(dNow), 2)
ss = Right("00" &Second(dNow), 2)

NewFolder = "c:\" &yy &mt &dd &hh &nn &ss

'Create the folder
objFSO.CreateFolder NewFolder
```

1.28 Extract data from HTML page